

UNA UTILIDAD EN FONETICA:
LA CARTA DE FORMANTES POR ORDENADOR

E. MARTINEZ CELDRAN

1. PROPOSITO.

Hace algún tiempo hicimos un programa para obtener las cartas de formantes de forma automática mediante el empleo del ordenador. Como creemos que es una herramienta útil para cualquier trabajo sobre vocales, ahora la ofrecemos para todos aquellos que la quieran aprovechar. El programa está en BASIC (GWBASIC) y corre en cualquier ordenador PC, XT/AT, con tarjeta gráfica CGA, EGA o VGA. Con unos mínimos conocimientos sobre BASIC, se puede copiar cada una de las líneas del programa y se tendrán miles de cartas con gran facilidad. El modelo de carta es el tradicional logarítmico, que representa en la ordenada el F1 y, en la abscisa, el F2 (Vid. A.Quilis, *Fonética acústica de la lengua española*, Madrid, Gredos, 1981, pp. 154-159).

2. EL FUNCIONAMIENTO DEL PROGRAMA.

Se ha imaginado que puede ser interesante combinar varios grupos de vocales, para comparar sistemas diferentes. También es posible que cada grupo represente una sola vocal, con sus distintas realizaciones, o que las vocales de cada hablante se consideren dentro del mismo grupo y se quieran comparar varios hablantes, etc. Por tanto, lo primero que requiere el programa es saber cuántos grupos de vocales se van a representar.

A continuación se pregunta si se desea que las vocales estén unidas con líneas. Si efectivamente queremos unir las, entonces es un requisito imprescindible que se introduzcan en el orden de las vocales cardinales. La carta se inspira en este modelo: las vocales cardinales. Esto quiere decir que si tuviéramos que representar las dieciséis vocales (ocho primarias y ocho secundarias) y unir las con líneas, tendríamos que introducir dos grupos y en orden: primero las primarias y, después, las secundarias. Si no queremos líneas, entonces el orden es indiferente. Cada grupo se distingue por un símbolo. Se ha previsto hasta un máximo de nueve símbolos diferentes.

Para la introducción de datos, se ofrecen dos subrutinas alternativas. Para aquellas personas que no sean muy duchos en el manejo de los ordenadores damos la posibilidad de entrar los datos directamente según lo va pidiendo el ordenador. En este caso, el programa demandará cuántas vocales va a tener el primer grupo; entonces, se van pidiendo los valores de primer formante y de segundo formante. Después de escribir cada respuesta, se ha de pulsar siempre la tecla "Intro". Y este proceso se repetirá para cada grupo. Para aquellas otras personas que estén acostumbradas al manejo de los ordenadores, ofrecemos otra subrutina diferente. Los datos pueden ser introducidos mediante un procesador de textos que produzca un fichero en código ASCII. La mayor parte de procesadores como el Word Perfect, por ejemplo, poseen esa posibilidad o mediante el procesador del PCTOOLS o el procesador KEDIT, etc.. Se ha de crear un fichero de tipo secuencial; es decir, cada línea contendrá exclusivamente los valores de primer formante y segundo, en ese orden y separados por una coma. Si se desean varios grupos de vocales, entonces los grupos irán en orden, sin separación alguna entre ellos.

La segunda subrutina ofrece varias ventajas sobre la primera. En el caso de entrar los datos por programa tiene la gran desventaja de que una vez pulsada la tecla "Intro" en una respuesta, ya no hay posibilidad de corrección; si el dato introducido es incorrecto, habrá que volver a comenzar. Y si se han de entrar treinta vocales, por

ejemplo, el proceso puede resultar bastante enojoso. Por otra parte, si la disposición de las vocales en grupo no es la adecuada, se tiene que volver a comenzar siempre de nuevo. En cambio, si se ha creado un fichero, los errores se corrigen directamente en el procesador. Se pueden cambiar los grupos formando bloques, fácilmente manejables con los procesadores, etc. Incluso, si se tienen programas de estadística que generan ficheros de datos, con una mínima manipulación pueden adaptarse para que los pueda leer este programa.

Una vez introducidos los datos, por alguna de las dos subrutinas, aparecerá en pantalla la carta. Si se desea imprimirla, se podrá hacer a través de la tecla "Print Screen" o "Imprimir Pantalla" (recuérdese que el programa GRAPHICS del MSDOS debe permanecer residente en memoria: por tanto, hay que lanzarlo previamente). Mucho mejor es tener algún programa que sea capaz de capturar pantallas gráficas y, luego, manipularlas en cuanto al tamaño. Esa posibilidad la ofrece, por ejemplo, el Word Perfect 5, a través del programa GRAB. Si se escribe GRAB/H, dentro del directorio WP, aparecerá una pantalla de ayuda que le enseñará cómo capturar el gráfico deseado. Es decir, se lanza el programa GRAB y aparece un mensaje donde se indica que el programa está instalado y permanece residente en memoria. Entonces se hace correr el programa de la carta de formantes (p.e.:GWBASIC CF). Una vez que tengamos la carta en pantalla, pulsando ALT-SHIFT-F9 a la vez se pondrá en marcha el programa GRAB. Aparece un rectángulo discontinuo que, con las flechas y SHIFT+flechas, puede adecuarse al tamaño del gráfico; cuando ya se tenga encuadrado el gráfico, se sale con F7. GRAB habrá creado un fichero, denominado GRAB.WPG, que contiene el gráfico capturado. Ya dentro del programa WP, la opción ALT-F9 y, en ella, la ilustración permiten introducir el gráfico, ponerle un pie a la figura y adecuar su tamaño a la medida deseada.

Tenemos todavía la carta en pantalla. Pulsando "Intro", se sale de ella y aparecen, entonces, los datos introducidos, por si queremos repasarlos o imprimirlos a través del "Prt Sc".

3. LAS CARTAS.

La carta más simple es la compuesta por un solo grupo; como ejemplo exponemos una que representa las cinco vocales del castellano (Fig.1).




Fig. 1.: Vocales castellanas sin contexto.

Cuando se quieren combinar dos sistemas vocálicos, hay que hacer dos grupos. Por ejemplo, la figura 2 es una carta que incluye las vocales del castellano dentro de las cardinales primarias (E. Martínez Celdrán, *Fonética*, Barcelona, Teide, 1984, p. 300)




Fig. 2.: Relación vocales cardinales-castellanas sin contexto.

Como se ve, la indicación sobre si se quieren líneas sirve para unir cada grupo de vocales; si se han entrado en el orden de las cardinales, las líneas no se interfieren.

Cuando se necesita representar una vocal neutra y se desea unir las demás con líneas, también se deben entrar los datos en dos grupos; por ejemplo, las vocales del catalán central tienen, además de las siete del triángulo, una neutra. Pues bien, el primer grupo tendrá siete vocales y el segundo sólo una; de esta forma quedan unidas por líneas las siete exteriores y la neutra queda aislada.

Esta disposición tiene el archivo de datos de la figura 3:




Fig. 3.:Vocales del catalán central.

250,2250
 400,1900
 550,1700
 700,1300
 550,1100
 400,900
 250,700
 450,1400

Como se ve, se trata de colocar los valores de ambos formantes en el orden adecuado y separados por comas.

Por último, cuando se pretende representar los campos de dispersión, entonces no son necesarias las líneas. En el caso de la figura 4, se presentan cuarenta vocales en total, repartidas en cuatro campos de dispersión, que podrán aparecer de la forma siguiente:

- 10 - [e]
- 10 - [ɛ]
- 10 - [ɔ]
- 10 - [o]

En una carta como ésta, se ve la dispersión en el sentido de que existen ees muy cerradas y muy abiertas; pero también hay unas cuantas que ocupan un terreno intermedio, que es difícil definir como cerrado o abierto; lo mismo sucede con las oes.




Fig. 4.: Dispersión de 40 vocales, repartidas en cuatro grupos.




Fig. 5.: Dispersión de 40 vocales, sin distribuir en grupos.

En la carta de la figura 5, se repiten las cuarenta vocales, pero se han introducido como si fuesen un solo grupo. Este sería el caso de no querer partir de diferencias previas y preestablecidas sobre qué vocales son las que aparecen.

La ilustración precedente puede ser modificada como se muestra en la figura 6: empezando por separar las vocales anteriores de las vocales posteriores.




Fig. 6.: Dispersión de 40 vocales, repartidas en dos grupos.

Por último, hemos desarrollado una máxima subdivisión: ocho grupos (Fig. 7). Con ello, quizá se vean mejor las áreas de intersección entre los campos de dispersión. Por otra parte, con esta carta se pueden contemplar hasta ocho símbolos diferentes.

Estas cartas son sólo una muestra de las posibilidades de este programa.




Fig. 7.: Dispersion de 40 vocales, repartidas en ocho grupos.

4. EL PROGRAMA.

A continuación se da su listado:

```

10 ' <<<<<<CARTA DE FORMANTES>>>>>>>>
20 KEY OFF: CLS
30 LOCATE 2, 20: COLOR 1, 7, 0: PRINT " CARTA DE
FORMANTES ": COLOR 14, 0
40 PRINT : PRINT : PRINT
50 INPUT "?Cuántos grupos de vocales hay? ", G
60 DIM F1(G, 100), F2(G, 100), V(100)
70 GOSUB 800
80 '=====AQUI SE DIBUJA LA CARTA DE FORMANTES
CON LOS DATOS=====
90 SCREEN 2
100 VIEW (10, 10)-(400, 170)

```

```

110 WINDOW (-3.5, -3)-(-2.55, -2.25)
120 FOR L = 1 TO G
130 FOR K = 1 TO V(L)
140 D = C: E = B
150 B = LOG(F1(L, K)) / LOG(10)
160 C = LOG(F2(L, K)) / LOG(10)
170 C = -C: B = -B
180 IF L = 1 THEN CIRCLE (C, B), 8.000001E-03
190 IF L = 2 THEN CIRCLE (C, B), 8.000001E-03:
LINE (C, B + .02)-(C, B)
200 IF L = 3 THEN CIRCLE (C, B), 8.000001E-03:
LINE (C, B)-(C, B - .02)
210 IF L = 4 THEN CIRCLE (C, B), 8.000001E-03:
LINE (C + .02, B)-(C, B)
220 IF L = 5 THEN CIRCLE (C, B), 8.000001E-03:
LINE (C, B)-(C - .02, B)
230 IF L = 6 THEN CIRCLE (C, B), 8.000001E-03:
LINE (C, B + .02)-(C, B - .02)
240 IF L = 7 THEN CIRCLE (C, B), 8.000001E-03:
LINE (C + .02, B)-(C - .02, B)
250 IF L = 8 THEN LINE (C, B + .01)-(C, B - .01)
260 IF L = 9 THEN LINE (C + .01, B)-(C - .01, B)
270 IF L > 9 THEN CIRCLE (C, B), .01
280 IF Q$ = "N" THEN 320
290 IF K = 1 THEN 310
300 LINE (D, E)-(C, B): GOTO 320
310 F = C: S = B
320 NEXT K
330 IF Q$ = "N" THEN 350
340 LINE (C, B)-(F, S)
350 NEXT L
360 FOR J = 500 TO 3500 STEP 500
370 C = LOG(J) / LOG(10)
380 C = -C
390 LINE (C, -3)-(C, -2.98)
400 NEXT J
410 LINE (-3.5, -2.99)-(-2.55, -2.99)
420 LINE (-2.56, -2.99)-(-2.56, -2.25)
430 LINE (-3.5, -2.25)-(-2.55, -2.25)
440 LINE (-3.5, -3)-(-3.5, -2.25)
450 FOR J = 200 TO 900 STEP 100
460 B = LOG(J) / LOG(10)
470 B = -B
480 LINE (-2.57, B)-(-2.25, B)
490 NEXT J
500 LOCATE 3, 54: PRINT "200": LOCATE 11, 54:
PRINT "400": LOCATE 16, 54: PRINT "600": LOCATE
19, 54: PRINT "800"

```

```

510 LOCATE 23, 42: PRINT "500": LOCATE 23, 16:
PRINT "1500"
520 LOCATE 23, 6: PRINT "2500"
530 FOR J = 750 TO 3000 STEP 500
540 C = LOG(J) / LOG(10): C = -C
550 CIRCLE (C, -3), .005
560 NEXT J
570 FOR H = 250 TO 1000 STEP 100
580 B = LOG(H) / LOG(10): B = -B
590 CIRCLE (-2.55, B), .005
600 NEXT H
610 K$ = INKEY$: IF K$ = "" THEN 610
620 SCREEN 0, 0, 0
630 '===== AQUÍ ES DONDE SE EXPONEN LOS DATOS
INTRODUCIDOS=====
640 COLOR 14, 0: PRINT "DATOS:"
650 TT = 0
660 FOR L = 1 TO G
670 PRINT " Grupo"; L
680 FOR K = 1 TO V(L)
690 TT = TT + 1: IF TT < 20 THEN LOCATE TT + 3, 1
700 IF TT > 20 THEN LOCATE TT - 19, 40
710 IF TT > 39 THEN INPUT " ", P$: CLS : TT = 0
720 PRINT K; F1(L, K); F2(L, K)
730 IF TT = V(L) THEN INPUT " ", P$
740 NEXT K
750 IF L <> G THEN CLS : TT = 0
760 NEXT L
770 CLS: INPUT "¿Deseas empezar de nuevo (S/N)? ",
S$
780 IF S$ = "S" THEN CLEAR : GOTO 10
790 CLS : SYSTEM : '----- FIN DEL PROGRAMA-----
800 '==SUBROUTINA PARA LA INTRODUCCION DE DATOS A
TRAVES DE UN FICHERO==
810 PRINT: INPUT "¿Deseas líneas que unan las
vocales (S/N)? ", Q$
820 PRINT : INPUT "¿Cómo se llama el fichero de
formantes? ", NOMFIC$
830 OPEN NOMFIC$ FOR INPUT AS #1
840 WHILE NOT EOF(1)
850 FOR L = 1 TO G
860 PRINT : PRINT "¿Cuántas vocales hay en el
grupo"; L; "?"; : INPUT " ", V(L)
870 FOR K = 1 TO V(L)
880 INPUT #1, FOR1, FOR2
890 F1(L, K) = FOR1: F2(L, K) = FOR2
900 NEXT K, L
910 WEND

```

```
920 CLOSE #1
930 RETURN
```

Los números que van desde el 800 hasta el 930 recogen la subrutina que recupera un archivo previamente establecido. Consideramos que esta es la forma óptima de trabajar. Pero en caso de resultar demasiado engorroso estar creando archivos con otros procesadores, se puede sustituir esta subrutina por la siguiente:

```
800 ' ===SUBROUTINA PARA LA INTRODUCCION DE DATOS
POR PROGRAMA===
810 PRINT : PRINT
820 PRINT "Si deseas líneas que unan las vocales,
deberás introducirlas en el orden de las vocales
cardinales, por grupo.": PRINT : INPUT "¿Lo deseas
(S/N) "; Q$
830 FOR L = 1 TO G
840 PRINT : PRINT "¿Cuántas vocales habrá en el
grupo"; L; : INPUT "? ", V(L)
850 CLS
860 FOR K = 1 TO V(L)
870 PRINT : PRINT "ENTRADA "; K; " DEL GRUPO "; L
880 PRINT
890 INPUT "FORMANTE PRIMERO "; F1(L, K)
900 INPUT "FORMANTE SEGUNDO "; F2(L, K)
910 NEXT K
920 NEXT L
930 RETURN
```

En la línea 60, se definen tres variables dimensionadas. El número 100 indica la cantidad de vocales que se pueden introducir en un solo grupo. Si en algún momento se prevé que van a ser más de 100 vocales en un grupo, este dato puede elevarse hasta que la memoria lo permita. El mismo usuario puede definir esta línea colocando los valores que suela utilizar.

Por otra parte, si se tiene un compilador, como el QUICKBASIC 4.0, este programa puede compilarse y hacerlo ejecutable directamente sin necesidad del apoyo del GWBASIC. También se gana un poco en velocidad, aunque este hecho es casi inapreciable, en este caso.